

Forritunarmál Hópverkefni 8

Ragnar Björn Ingvarsson, rbi3
Daníel Snær Halldórsson, dsh11
Ólafur Sær Sigursteinsson, oss27

16. október 2024

1

```
;;; Notkun: y = sort(x);
;;; Fyrir: x er listi talna.
;;; Eftir: y er listi sem inniheldur sömu gildi
;;;        og x en í vaxandi röð.
rec fun sort(x)
{
  ;;; Notkun: split(x,&y,&z);
  ;;; Fyrir: x er listi.
  ;;; Eftir: y og z innihalda lista þannig að
  ;;;        samskeyting þeirra inniheldur öll
  ;;;        gildi x, en e.t.v. í annarri röð.
  ;;;        Mismunur lengda y og z er í mesta
  ;;;        lagi einn.
  rec fun split(x,&y,&z)
  {
    if ( x != [] )
    {
      y = head(x) : y;
      split(tail(x), &z, &y);
    };
  };

  ;;; Notkun: z = merge(x,y);
  ;;; Fyrir: x og y eru listar sem innihalda
  ;;;        tölur.
  ;;;        Báðir listarnir eru í vaxandi röð
  ;;; Eftir: z inniheldur öll gildin úr x og y
  ;;;        í vaxandi röð.
  rec fun merge(x,y)
  {
    if (x == []) { return y };
    if (y == []) { return x };
    if (head(x) < head(y))
    {
      return head(x) : merge(tail(x), y);
    };
    return head(y) : merge(x, tail(y));
  };

  if (x == [] || tail(x) == []) { return x };
  var y = [], z = [];
  split(x, &y, &z);
  return merge(sort(y), sort(z));
};
```

```
1      writeln(sort([1,5,3,2,45,2,6]));
2      writeln(sort([]));
3  };
4
:!java -jar morpho.jar msort
[1,2,2,3,5,6,45]
null
```

Ath. að þegar morpho prentar tóma listann prentast "null"svo sort([]) skilar í raun tóma listanum [].